

ANALITIČKA GEOMETRIJA RAVNINE RAČUNALOM


MAJA BATOR, ZVONKO ČERIN I MILENA ČULAV, ZAGREB

SAŽETAK. Opisuje se kako se u programu Maple V programiraju osnovne procedure za analitičku geometriju ravnine u pravokutnim koordinatama. Kao ilustraciju mogućih primjena dana su rješenja nekoliko zadataka iz zbirke za prvi razred gimnazije.

U ovom članku opisujemo kako se uz pomoć računala može obraditi analitička geometrija ravnine. Pritom se koristimo složenim programom **Maple V** koji je po svojim mogućnostima sličan programu **Mathematica**. To su sigurno najpoznatiji i najrašireniji sustavi koji "znaju matematiku". Svaki od njih ima vlastiti programski jezik i naš zadatak svodi se na to da na tom jeziku opišemo nove funkcije i postupke (eng. procedures) koje trebamo za rješavanje geometrijskih zadataka analitičkom metodom.

Ideja analitičke geometrije je da se geometrijska svojstva tvorevina opišu algebarskim izrazima te da se ona prouče metodama algebre. To se najjednostavnije postiže u ravnini pa ćemo od nje i krenuti.

Jer je **točka** osnovna tvorevina u geometriji ravnine, moramo se prvo odlučiti kako ih prikazati algebarski. To se najlakše ostvaruje odabirom koordinatnog sustava. Poznato je više takvih sustava (npr. pravokutni, kosokutni, polarni, trilinearni, baricentrički,...), ali daleko najprirodniji je pravokutni koordinatni sustav pa ćemo ga i mi odabrati.



SLIKA 1. Pravokutni koordinatni sustav.

Taj sustav je zadan ako odaberemo dva orijentirana međusobno okomita pravca p_x i p_y i ako na jednom od njih (recimo na p_x) odaberemo točku E različitu od točke njihovog presjeka $O = p_x \cap p_y$. Točku O nazivamo *ishodište*, pravci p_x i p_y su x -os i y -os, a E je *jedinična točka* (vidi sliku 1). Položaj bilo koje točke T u ravni jednoznačno je određen uređenim parom (x, y) realnih brojeva koji se zovu (pravokutne) *koordinate* točke T . Brojevi x i y su sljedeći omjeri orijentiranih dužina

$$x = \frac{|OT_x|}{|OE|}, \quad y = \frac{|OT_y|}{|OE|},$$


gdje su T_x i T_y projekcije točke T na x -os i na y -os.

Zadavanje točaka u ravni u programu Maple V je isto tako jednostavno budući da je točka zadana uređenim parom realnih brojeva. Na primjer, naredbom

`tA:=[2, 3]: tB:=[5, 7]: tC:=[-2, 0]: tT:=[x, y]:`

zadali smo četiri točke ravni $A(2, 3)$, $B(5, 7)$, $C(-2, 0)$, $T(x, y)$.

Sada kada znamo zadavati točke ravni opišimo kako se računa *udaljenost* među njima.



SLIKA 2. Udaljenost dvije točke $A(a, u)$ i $B(b, v)$ u ravni je $|AB| = \sqrt{(b - a)^2 + (v - u)^2}$.

Ako su A i B dvije točke u ravni s koordinatama (a, u) i (b, v) vidimo da je udaljenost $|AB|$ hipotenuza pravokutnog trokuta ABC , gdje je C sjecište paralele s x -osi točkom A i paralele s y -osi točkom B . Katete tog trokuta su (horizontalna) $|b - a|$ i (vertikalna) $|u - v|$ (vidi sliku 2). Zato je, prema Pitagorinom teoremu, ta udaljenost jednaka

$$|AB| = \sqrt{(b - a)^2 + (v - u)^2}.$$

U Maple-u V udaljenost se računa sljedećom naredbom:

`udaljenost2t := proc(A,B) local a,u,b,v;`


```
a := A[1] : u := A[2] : b := B[1] : v := B[2] :
factor(simplify(sqrt((b-a)^2+(v-u)^2))): end:
```

Za ime te funkcije odabrali smo riječ *udaljenost2t* (skraćenica za "udaljenost dvije točke"). Za njezinu definiciju koristimo osnovnu naredbu *proc*. Njene varijable su dvije točke (tj. dva uređena para realnih brojeva A i B). Ona prvo odredi prvu i drugu komponentu svake točke i spremi ih kao lokalne varijable a , u , b i v . U sljedećem koraku izračuna drugi korijen (naredba *sqr*t) iz izraza $(b-a)^2 + (v-u)^2$ i onda pokušava što je više moguće pojednostavniti taj izraz (naredba *simplify*) i još rastaviti na faktore (naredba *factor*). Pritom naredba *end* označava završetak djelovanja funkcije.

Da bi smanjili tipkanje uvodimo skraćenicu *FS* za istovremenu upotrebu naredbi *factor* i *simplify* koje ćemo često koristiti.

```
FS := proc(u) factor(simplify(u)): end:
```

U mnogim primjenama važno je odrediti *polovište dužine* čiji krajevi su poznati. Iz priložene slike 3 se vidi da su koordinate polovišta P dužine s krajevima u točkama $A(a, u)$ i $B(b, v)$ jednake $\frac{a+b}{2}$ i $\frac{u+v}{2}$.



SLIKA 3. Koordinate točke polovišta P dužine \overline{AB} su aritmetičke sredine koordinata krajeva.

Postupak za traženje polovišta dužine u Maple-u V izgleda ovako:

```
poloviste := proc(A,B) local a,u,b,v;
a := A[1] : u := A[2] : b := B[1] : v := B[2] :
FS([(a+b)/2, (u+v)/2]): end:
```

Nešto složenije je naći koordinate točke koja dijeli dužinu u zadanom omjeru. Dakle, tražimo točku C na pravcu AB takvu da je omjer duljina orijentiranih dužina $\frac{|AC|}{|CB|}$ jednak zadanom broju λ (koji mora

biti različit od -1). Njene koordinate su $\frac{a + \lambda b}{1 + \lambda}$ (apscisa) i $\frac{u + \lambda v}{1 + \lambda}$ (ordinata). Najčešće se λ zadaje kao kvocijent $\frac{m}{n}$ dvaju cijelih brojeva m i n (koji nisu jedan drugome suprotni).

Pripadne procedure u Maple-u V su:

```
omjer_lambda := proc(A,B,k) local a,u,b,v;
    a := A[1] : u := A[2] : b := B[1] : v := B[2] :
    FS([(a+k*b)/(1+k), (u+k*v)/(1+k)]): end;

omjer_mn := proc(A,B,m,n) local a,u,b,v;
    a := A[1] : u := A[2] : b := B[1] : v := B[2] :
    FS([(a*n+b*m)/(m+n), (u*n+v*m)/(m+n)]): end;
```

Polovište je, naravno, specijalan slučaj prethodnog postupka. Dobivamo ga iz općeg slučaja za $\lambda = 1$ odnosno $m = 1$ i $n = 1$.

Pored točaka u ravnini su najvažniji **pravci**. Prisjetimo se da se pravci zadaju linearnim jednadžbama. Točnije, ako su a , b i c realni brojevi takvi da a i b nisu istovremeno nula, onda je skup svih točaka $P(x, y)$ čije koordinate zadovoljavaju jednadžbu

$$ax + by + c = 0$$

pravac. Obrnuto, za svaki pravac mogu se naći takvi brojevi a , b i c da koordinate svake točke tog pravca zadovoljavaju gornju jednadžbu.

Zato ćemo u Maple-u V pravce zadavati uređenom trojkom $[a, b, c]$ koeficijenata njegove linearne jednadžbe. Na primjer, naredbom

```
pX:=[1, 0, 0] : pY:=[0, 1, 0] : pD:=[1, -1, 0] : pG:=[-1, 2, 2] :
```

zadali smo četiri pravca ravnine, i to x -os, y -os, simetralu prvog i trećeg kvadranta i pravac $-x + 2y + 2 = 0$.

Pravac se obično zadaje jednom svojom točkom $P_0(x_0, y_0)$ i tangansom k kuta kojeg zatvara s pozitivnim smjerom x -osi. Njegova jednadžba je

$$y - y_0 = k(x - x_0),$$

pa ga u Maple-u V zadajemo na sljedeći način:

```
pravackt := proc(k, b) local b1, b2; b1 := b[1] : b2 := b[2] :
    FS([k, -1, b2-b1*k]): end proc;
```

Drugi najčešći način zadavanja pravca je pomoću njegove dvije različite točke $P_1(x_1, y_1)$ i $P_2(x_2, y_2)$. Njegova jednadžba je

$$y - y_1 = \frac{y_2 - y_1}{x_2 - x_1} (x - x_1)$$

ako je $x_1 \neq x_2$. Prepišemo li je u obliku

$$(y_2 - y_1)x + (x_1 - x_2)y + (x_1 y_2 - x_2 y_1) = 0$$

vidimo da vrijedi i ako je $x_1 = x_2$. Zato u ovom slučaju program u Maple-u V izgleda ovako:

```
pravac2t:= proc(a,b) local a1, a2, b1, b2;
             a1:=a[1]: a2:=a[2]: b1:=b[1]: b2:=b[2]:
             FS([a2-b2,b1-a1,a1*b2-b1*a2]):end:
```

Ponekad je korisna i sljedeća procedura kojom ispituje li dana točka na zadanom pravcu. Slovo Q u njenom imenu dolazi od engleske riječi "question" za pitanje.

```
točka_na_pravcuQ:=proc(t,p) local a,b,x,y,z,iz; a:=t[1]:
b:=t[2]:x:=p[1]:y:=p[2]:z:=p[3]:iz:=FS(a*x+y*b+z):
if iz=0 then print('Točka je na pravcu!!!'): else
print('Točka je na pravcu akko je ovaj izraz jednak 0.'):
print(iz):fi:end:
```

Sada lagano možemo naći uvjet koji koordinate tri točke $P_1(x_1, y_1)$, $P_2(x_2, y_2)$ i $P_3(x_3, y_3)$ moraju zadovoljavati da bi one ležale na istom pravcu (tj. da su kolinearne). Naime, gore smo vidjeli kako izgleda jednadžba pravca kroz točke P_1 i P_2 . Budući da točka P_3 mora ležati upravo na tom pravcu njezine koordinate moraju zadovoljavati tu jednadžbu. Drugim riječima mora vrijediti

$$(y_2 - y_1)x_3 + (x_1 - x_2)y_3 + (x_1y_2 - x_2y_1) = 0.$$

To nam daje ideju kako definirati proceduru u Maple-u V koja će određivati da li su tri zadane točke kolinearne ili ne.

```
kolinearneQ:=proc(a,b,c) local a1,a2,b1,b2,c1,c2,iz;
a1:=a[1]:a2:=a[2]:b1:=b[1]:b2:=b[2]:c1:=c[1]:c2:=c[2]:
iz:=FS(a1*b2-a1*c2-b1*a2+b1*c2+c1*a2-c1*b2):
if iz=0 then print('Tocke su kolinearne!!!'):
else print('Tocke ce biti kolinearne ako i samo
ako je sljedeci izraz jednak nuli'): print(iz):fi:end:
```

Dva pravca $A_1x + B_1y + C_1 = 0$ i $A_2x + B_2y + C_2 = 0$ u ravnini se ili sijeku ili su paralelni. Ako se sijeku, onda su koordinate presjeka dane izrazima

$$x = \frac{B_1 C_2 - B_2 C_1}{A_1 B_2 - A_2 B_1}, \quad y = \frac{C_1 A_2 - C_2 A_1}{A_1 B_2 - A_2 B_1}.$$

Pretpostavka da se zadani pravci sijeku osigurava da su nazivnici u tim izrazima različiti od nule. Paralelni pravci nemaju presjek (u konačnoj ravnini) i $A_1 B_2 - A_2 B_1 = 0$ je uvjet na njihove koeficijente koji mora biti ispunjen da bi oni bili paralelni. Prema tome, u Maple-u V sljedećom naredbom određujemo presjek dva pravca ili otkrivamo da su oni paralelni.

```
presjek2p := proc(x,y) local a,b,c,i,j,k;
a:=x[1]:b:=x[2]:c:=x[3]:i:=y[1]:j:=y[2]:k:=y[3]:
if FS(-i*b+a*j)<>0 then
FS([( -j*c+k*b)/(-i*b+a*j), -(i*c+a*k)/(-i*b+a*j)]):
```

```
else print('Pravci su paralelni!!!'):fi:end:
```


Ako se pravci p_1 i p_2 s jednadžbama

$$A_1 x + B_1 y + C_1 = 0 \quad \text{i} \quad A_2 x + B_2 y + C_2 = 0$$

sijeku onda se tangens kuta φ među njima određuje formulom

$$\tan \varphi = \frac{A_1 B_2 - A_2 B_1}{A_1 A_2 + B_1 B_2} = \frac{k_2 - k_1}{1 + k_1 k_2},$$

gdje su $k_1 = -\frac{A_1}{B_1}$ i $k_2 = -\frac{A_2}{B_2}$ tangensi kuteva φ_1 i φ_2 što ih oni čine s x -osi (vidi sliku 4).



SLIKA 4. Kut između pravaca koji se sijeku.

Sljedeća procedura to ostvaruje u Maple-u V:

```
tankut2p:=proc(x,y) local a,b,c,i,j,k;
  a:=x[1]:b:=x[2]:c:=x[3]:i:=y[1]:j:=y[2]:k:=y[3]:
  if FS(a*i+b*j)<>0 then FS((a*j-i*b)/(a*i+b*j)):
  else print('Pravci su okomiti!!!'):fi:end:
```

Sada vidimo da su pravci p_1 i p_2 paralelni ako je kut između njih jednak nuli ili ako je $A_1 B_2 - A_2 B_1 = 0$, odnosno $k_1 = k_2$. S druge strane oni su okomiti ako je $A_1 A_2 + B_1 B_2 = 0$, odnosno $k_2 = -\frac{1}{k_1}$ (nazivnik gornjeg razlomka jednak je nula pa je kut pravi).

Te primjedbe opravdavaju sljedeće definicije procedura koje traže paralelu danom pravcu kroz danu točku i okomicu na dani pravac kroz danu točku. Njima slične su one koje istražuju da li su dva pravca paralelna ili okomita.

```
paralela:=proc(t,p) local a,b,x,y;
```

```

a:=t[1]:b:=t[2]:x:=p[1]:y:=p[2]:
FS([x,y,-x*a-b*y]):end:

okomica:=proc(t,p) local a,b,x,y;
a:=t[1]:b:=t[2]:x:=p[1]:y:=p[2]:
FS([y,-x,x*b-y*a]):end:

paralelniQ:=proc(p,q)local a,b,x,y,iz:
x:=p[1]:y:=p[2]:a:=q[1]:b:=q[2]:iz:=FS(a*y-x*b):
if iz=0 then print('Pravci su paralelni!!!'): else
print('Pravci su paralelni akko je sljedeci izraz
jednak nuli'): print(iz):fi:end:

okomitiQ:=proc(p,q)local a,b,x,y,iz:
x:=p[1]:y:=p[2]:a:=q[1]:b:=q[2]:iz:=FS(a*x+y*b):
if iz=0 then print('Pravci su okomiti!!!'): else
print('Pravci su okomiti akko je sljedeci izraz
jednak nuli'): print(iz):fi:end:

```

Za tri različita pravca p_1 , p_2 i p_3 s jednadžbama

$$A_1 x + B_1 y + C_1 = 0, \quad A_2 x + B_2 y + C_2 = 0, \quad A_3 x + B_3 y + C_3 = 0$$

idemo odrediti uvjet što ga njihovi koeficijenti moraju zadovoljavati da bi oni bili kopunktalni ili konkurentni (tj. da su paralelni ili da se sijeku u jednoj točki).

Ako se pravci p_1 i p_2 sijeku u točki $T(x, y)$ onda znamo da su x i y kvocijenti $\frac{B_1 C_2 - B_2 C_1}{A_1 B_2 - A_2 B_1}$ i $\frac{C_1 A_2 - C_2 A_1}{A_1 B_2 - A_2 B_1}$. Uvrstimo li ih u p_3 dobivamo

$$\frac{A_3 B_1 C_2 - A_3 B_2 C_1 + B_3 C_1 A_2 - B_3 C_2 A_1 + C_3 A_1 B_2 - C_3 A_2 B_1}{A_1 B_2 - A_2 B_1} = 0.$$

Dakle, ti pravci prolaze istom točkom jedino ako je izraz u brojniku

$$\Delta = A_3 (B_1 C_2 - B_2 C_1) + B_3 (C_1 A_2 - C_2 A_1) + C_3 (A_1 B_2 - A_2 B_1)$$

jednak nuli.

Ako su pravci p_1 , p_2 i p_3 paralelni onda su njihovi koeficijenti vezani relacijama $A_2 = \lambda A_1$, $B_2 = \lambda B_1$, $A_3 = \mu A_1$, $B_3 = \mu B_1$, za neke realne brojeve λ i μ različite od nule. Uvrstimo li te vrijednosti u Δ opet se dobiva nula.

Obrnuto, ako je $\Delta = 0$, podijelimo li tu jednakost s $A_1 B_2 - A_2 B_1$ vidimo da presjek $T\left(\frac{B_1 C_2 - B_2 C_1}{A_1 B_2 - A_2 B_1}, \frac{C_1 A_2 - C_2 A_1}{A_1 B_2 - A_2 B_1}\right)$ pravaca p_1 i p_2 zadovoljava jednadžbu od p_3 . Dakle, ta tri pravca su kopunktalna. To smo mogli načiniti jedino ako je $A_1 B_2 - A_2 B_1 \neq 0$, tj. ako p_1 i p_2 nisu paralelni.


Ako su p_1 i p_2 paralelni onda postoji realan broj λ takav da je $A_2 = \lambda A_1$ i $B_2 = \lambda B_1$. Uvrstimo li to u izraz $\Delta = 0$ dobijemo

$$(A_1 B_3 - A_3 B_1)(\lambda C_1 - C_2) = 0.$$

U produktu s lijeve strane druga zagrada nije nula jer je $p_1 \neq p_2$. Dakle, $A_1B_3 - A_3B_1 = 0$ što znači da su p_1 i p_3 također paralelni i dokaz je potpun jer su onda p_1 , p_2 i p_3 paralelni.

Tako smo objasnili kako u Maple-u V definirati proceduru kojom ispitujemo da li su tri zadana pravca konkurentna.

```
konkurentniQ:=proc(x,y,z) local a,b,c,i,j,k,p,q,r,iz;
  a:=x[1]:b:=x[2]:c:=x[3]:i:=y[1]:j:=y[2]:
  k:=y[3]:p:=z[1]:q:=z[2]:r:=z[3]:
  iz:=FS(a*j*r-a*k*q-i*b*r+i*c*q+p*b*k-p*c*j):
  if iz=0 then print('Pravci su konkurentni!!!'): else
  print('Pravci ce biti konkurentni akko je izraz'):
  print(iz): print('jednak nuli'):fi:end:
```



SLIKA 5. Refleksije točke u odnosu na točku (lijevo) i refleksija točke u odnosu na pravac (desno).

Pri rješavanju zadataka analitičke geometrije često treba odrediti projekciju točke na pravac. Ta projekcija je presjek tog pravca i okomice na njega zadanom točkom. Dakle ako napišemo

```
Q:=presjek2p(m, okomica(P, m));
```

Maple V će nam izračunati koordinate projekcije Q točke P na pravac m . Zato pripadna procedura izgleda ovako:

```
projekcija:=proc(P,m) local p,q,a,b,c;
  p:=P[1]:q:=P[2]:a:=m[1]:b:=m[2]:c:=m[3]:
  FS([- (c*a+b*q*a-p*b^2)/(b^2+a^2),
  (-b*c+q*a^2-a*p*b)/(b^2+a^2)]):end:
```

Slično, mnogo puta trebamo naći refleksije točke u odnosu na neku točku ili neki pravac. Kada točku P reflektiramo u odnosu na točku Q koordinate reflektirane točke R mogu se odrediti ranije definiranom

funkcijom `omjer_lambda` za $\lambda = -2$ jer točka R dijeli dužinu \overline{PQ} u omjeru -2 .

```
Q:=omjer_lambda(P, Q, -2);
```

Zato je funkcija koja određuje reflektiranu točku R u Maple-u V dana ovako:

```
refleksija_t:=proc(P,Q) local p,q,a,b;
  p:=P[1]:q:=P[2]:a:=Q[1]:b:=Q[2]:
  FS([2*a-p, 2*b-q]):end;
```

Slučaj refleksije točke P obzirom na pravac m je sličan. Sada je tražena točka R refleksija točke P u odnosu na projekciju Q točke P na pravac m .

```
R:=omjer_lambda(P, projekcija(P, m), -2);
```

Prema tome pripadna procedura ima sljedeći oblik:

```
refleksija_p:=proc(P,m) local p,q,a,b;
  p:=P[1]:q:=P[2]:a:=m[1]:b:=m[2]:c:=m[3]:
  FS([(b^2*p-a^2*p-2*a*b*q-2*c*a)/(a^2+b^2),
    (a^2*q-b^2*q-2*a*b*p-2*c*b)/(a^2+b^2)]):end;
```

Sada kada smo definirali osnovne funkcije analitičke geometrije ravnine, na nekoliko primjera ćemo ilustrirati kako se one upotrebljavaju za rješavanje zadataka. Naš prvi primjer je zadatak 395. u knjizi [1] koji glasi ovako:

Zadatak 1. Dokaži da je površina P trokuta ABC s vrhovima u točkama $A(x_1, y_1)$, $B(x_2, y_2)$ i $C(x_3, y_3)$ dana formulom:

$$P = \frac{|x_1(y_2 - y_3) + x_2(y_3 - y_1) + x_3(y_1 - y_2)|}{2}$$

odnosno


$$P = \frac{|y_1(x_2 - x_3) + y_2(x_3 - x_1) + y_3(x_1 - x_2)|}{2}.$$

Rješenje. Znamo da je površina trokuta jednaka polovici umnoška duljina bilo koje stranice i njoj pripadne visine. Pomoću procedura u Maple-u V koje smo ranije definirali lagano izračunamo:

```
A:=[x[1],y[1]]; B:=[x[2],y[2]]; C:=[x[3],y[3]];
D:=projekcija(C,pravac2t(A,B));
P:=FS(udaljenost2t(A,B)*udaljenost2t(C,D)/2);
```

Računalo će nam izračunati

$$\frac{1}{2} \sqrt{\frac{(-y_3x_1 + x_3y_1 + y_3x_2 - x_2y_1 - x_3y_2 + x_1y_2)^2}{x_2^2 - 2x_2x_1 + x_1^2 + y_1^2 - 2y_1y_2 + y_2^2}} \sqrt{x_2^2 - 2x_2x_1 + x_1^2 + y_1^2 - 2y_1y_2 + y_2^2}.$$



SLIKA 6. Površina trokuta ABC je polovica umnoška duljina osnovice \overline{AB} i pripadne visine \overline{CD} .

Računalo je ipak samo stroj pa on ne skрати nazivnik u prvom kvadratnom korijenu s drugim kvadratnim korijenom (premda su oni očito jednaki), a isto tako ne primjećuje da je drugi korijen kvadrata u brojniku jednak apsolutnoj vrijednosti

$$|-y_3x_1 + x_3y_1 + y_3x_2 - x_2y_1 - x_3y_2 + x_1y_2|.$$

Kad načinimo ta skraćivanja očigledno dobijemo traženu formulu. \square

Interesantno je da bez znaka apsolutne vrijednosti gornja formula daje izraz za orijentiranu površinu trokuta ABC . Ako je taj trokut pozitivno orijentiran, tj. ako je obilazak $ABCA$ u smjeru suprotnom od smjera kazaljke na satu, onda će taj broj biti pozitivan, a inače je negativan. Postupak za određivanje te, tzv. orijentirane površine trokuta, je u Maple-u V ostvaren na sljedeći način:


```
povrsina3t:=proc(A,B,C) local a,x,b,y,c,z;
a:=A[1]:x:=A[2]:b:=B[1]:y:=B[2]:c:=C[1]:z:=C[2]:
FS((x*c-b*x-a*z+a*y+b*z-c*y)/2): end;
```

Naš drugi primjer je zadatak 425 iz iste knjige [1].

Zadatak 2. Neka je ABC trokut i P_1, P_2, P_3 polovišta stranica \overline{BC} , \overline{CA} i \overline{AB} . Dužine $\overline{AP_1}, \overline{BP_2}$ i $\overline{CP_3}$ zovemo **težišnicama** trokuta ABC . Dokaži analitički da se sve tri težišnice sijeku u jednoj točki koju zovemo **težištem** trokuta i da težište dijeli svaku od težišnica u omjeru 2 : 1 računajući od vrha.

Rješenje. Dokaz na upaljenom računalu, u Maple-u V, počinje utipkavanjem sljedećeg:

```
A:=[x[1],y[1]]; B:=[x[2],y[2]]; C:=[x[3],y[3]];
```



SLIKA 7. Trokut ABC i njegove težišnice $\overline{AP_1}$, $\overline{BP_2}$ i $\overline{CP_3}$ koje se sijeku u težištu T .

```
D:=poloviste(B,C); E:=poloviste(C,A); F:=poloviste(A,B);
konkurentniQ(pravac2t(A,D),pravac2t(B,E),pravac2t(C,F));
```

U nestvarno brzom vremenu računalo će ispisati vrijednost nula što pokazuje da se težišnice sijeku u jednoj točki. O kojoj se točki radi otkrivamo naredbom:

```
T:=presjek2p(pravac2t(A,D), pravac2t(B,E));
```

Točka T ima koordinate $(\frac{x_1+x_2+x_3}{3}, \frac{y_1+y_2+y_3}{3})$ pa odmah možemo napisati proceduru u Maple-u V koja trokutu pridružuje njegovo težište.

```
teziste:=proc(A,B,C) local a,x,b,y,c,z;
    a:=A[1]:x:=A[2]:b:=B[1]:y:=B[2]:c:=C[1]:z:=C[2]:
    FS([(a+b+c)/3, (x+y+z)/3]):end;
```

Da bismo dokazali drugu tvrdnju zadatka mi ćemo odrediti točku koja dijeli težišnicu vrha A (tj. dužinu \overline{AD}) u omjeru $2:1$ računajući od vrha A i pokazati da se ona podudara s točkom T (težištem trokuta ABC). Isti argument može se ponoviti i za težišnice vrhova B i C .


```
G:=omjer_mn(A,D,2,1): udaljenost2t(G,T);
```

Budući da je dobivena vrijednost nula, točke G i T se podudaraju, pa je time zadatak riješen. \square

Treći primjer je zadatak 989 također iz knjige [1].

Zadatak 3. Dokaži da polovišta stranica trokuta i nožišta visina leže na istoj kružnici.

Rješenje. Bez smanjenja općenitosti možemo pretpostaviti da su točke A , B i C odabrane u ravnini tako da su im koordinate $(0, 0)$, $(c, 0)$ i (u, v) , gdje su c , u i v realni brojevi, pri čemu c i v nisu jednaki nuli.



SLIKA 8. Polovišta stranica i nožišta visina leže na istoj kružnici.

$eA := [0, 0] : eB := [0, c] : eC := [u, v] :$

Zatim odredimo polovišta stranica primjenom procedure `poloviste`:

```
eAp:=poloviste(eB,eC); eBp:=poloviste(eC,eA);
eCp:=poloviste(eA,eB);
```

Nožišta visina su projekcije vrhova na nasuprotne pravce stranica:

```
eApp:=projekcija(eA,pravac2t(eB,eC));
eBpp:=projekcija(eB,pravac2t(eC,eA));
eCpp:=projekcija(eC,pravac2t(eA,eB));
```

Središte trokutu opisane kružnice je točka presjeka simetrala njegovih stranica. Dakle, u našem slučaju je središte S trokuta $A'B'C'$ s vrhovima u polovištima stranica.

```
eS:=presjek2p(okomica(eAp,pravac2t(eBp,eCp)),
               okomica(eBp,pravac2t(eCp,eAp)));
```

Primijenimo li isti postupak na trokut $A''B''C''$ s vrhovima u nožištima visina možemo odrediti središte T njemu opisane kružnice.

```
eT:=presjek2p(okomica(eApp,pravac2t(eBpp,eCpp)),
               okomica(eBpp,pravac2t(eCpp,eApp)));
```

Nakon utipkavanja prethodnih naredbi računalo nam ispisuje koordinate točaka S i T . Vidi se da su one jednake, pa su zato S i T jedna te ista točka.

Da bi dokazali tvrdnju zadatka moramo još pokazati da su radijusi opisanih kružnica trokuta $A'B'C'$ i $A''B''C''$ jednaki.

```
FS(udaljenost2t(eS,eCp)-udaljenost2t(eT,eCp));
```

Izračunata vrijednost je nula, čime je dokaz zadatka završen.

Interesantno je da možemo naći da je radijus gornje kružnice točno jednak polovici radijusa trokutu ABC opisane kružnice. Da to provjerimo koristeći gornju metodu prvo odredimo koordinate središta O te kružnice i zatražimo od Maple-a V da izračuna sljedeće:

$\text{FS}(\text{udaljenost2t}(\mathbf{e0}, \mathbf{eC}) / \text{udaljenost2t}(\mathbf{eS}, \mathbf{eCp}))$;

Naravno, rezultat je dva. □

REFERENCE

- [1] Boris Pavković i Darko Veljan, *Matematika 1 – Zbirka zadataka s uputama i rješenjima za prvi razred srednjih škola (14. dopunjenjo izdanje)*, Školska knjiga, Zagreb, 1999.

ODRANSKA 8, 10000 ZAGREB, HRVATSKA, EUROPA
E-mail address: batormaja@yahoo.com

KOPERNIKOVA 7, 10020 ZAGREB, HRVATSKA, EUROPA
E-mail address: CERIN@MATH.HR

LHOTKINA 3, 10000 ZAGREB, HRVATSKA, EUROPA
E-mail address: mickey@student.math.hr